# Maintainer:

pkgname=perl-file-path-tiny
pkgver=1.0
pkgrel=2
pkgdesc='recursive versions of mkdir() and rmdir() without as much overhead as File::Path'
arch=('any')
url="https://metacpan.org/release/File-Path-Tiny"
license=('PerlArtistic' 'GPL')
depends=('perl')
checkdepends=('perl-test-exception')
options=('!emptydirs')
source=("https://search.cpan.org/CPAN/authors/id/D/DM/DMUEY/File-Path-Tiny-$pkgver.tar.gz")
sha256sums=('2ec178da3b9899e4b466ab8b71edbb2bf23a0307ebe02fec7aa1f5826f61f55a')

build() {
  cd File-Path-Tiny-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd File-Path-Tiny-$pkgver
  make test
}

package() {
  cd File-Path-Tiny-$pkgver
  make install DESTDIR="$pkgdir"
}
